

THE ADX

The calculation of the average directional movement (Adx) indicator is built on the intuitive notion that a trend is a series of price ranges extending in a consistent direction. In sidebar Figure 1, example A, the second day's trading range is higher than the first day's trading range, indicating positive directional movement. In example B, the second day's trading range is below the first day's trading range, an indication of negative directional movement. Example C is more complicated because the second day's range is both lower and higher than the first day's range. Directional movement is only considered to be up, down or not present. Therefore, the larger part of the day's range extending beyond the previous day's range is used to identify directional movement. In example C, the largest part of the second day is higher; consequently, the directional movement is positive. In example D, the largest part of the second day's range is lower so that the directional movement is negative. In example E, the second day's range is within the first day's range so the directional movement is zero.

Directional movement for the ADX is expressed as a function of true range (TR). True range is the largest of the following:

- 1 The difference between today's high and today's low.
- 2 The difference between today's high and yesterday's close.
- 3 The difference between today's low and yesterday's close.

In the Excel 4.0 spreadsheet (sidebar Figure 2), the first calculation for Adx is the true range value. This is performed in column E. The formula for cell E3 is:

$$=MAX(B3-C3,ABS(B3-D2),ABS(C3-D2))$$

Next, column F determines the positive directional movement or returns zero if there is no positive directional movement. The formula for cell F3 is:

$$=IF(B3-B2>C2-C3,MAX(B3-B2,0),0)$$

Column G calculates the negative directional movement or returns zero if there is no negative directional movement. The formula for cell G3 is:

$$=IF(C2-C3>B3-B2,MAX(C2-C3,0),0)$$

The daily calculations are volatile and so the data needs to be smoothed. First, sum the last 14 periods for TR, +DM and -DM. The formula for summing the TR is in cell H16:

$$=SUM(E3:E16)$$


The formula for summing the +DM is in cell I16:

$$=SUM(F3:F16)$$

The formula for summing the -DM is in cell J16:

$$=SUM(G3:G16)$$

DIRECTIONAL MOVEMENT EXAMPLES



SIDEBAR FIGURE 1: In examples A and C, the directional movement for day 2 compared with day 1 is positive because the largest portion of day 2's range is above day 1. In examples B and D, the directional movement for day 2 compared with day 1 is negative because the largest portion of day 2's range is below day 1. In example E, day 2 has zero directional movement compared with day 1.

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P
1	Date	High	Low	Close	TR	+DM	-DM	TR14	+DM14	-DM14	+DI14	-DI14	DI	DI	DX	ADX
2	920102	417.27	411.04	417.26												
3	920103	419.79	416.16	419.34	3.63	2.52	0.00									
4	920106	419.44	416.92	417.96	2.52	0.00	0.00									
5	920107	417.96	415.20	417.40	2.76	0.00	1.72									
6	920108	420.23	415.02	418.10	5.21	2.27	0.00									
7	920109	420.50	415.85	417.61	4.65	0.27	0.00									
8	920110	417.62	413.31	415.10	4.31	0.00	2.54									
9	920113	415.36	413.54	414.34	1.82	0.00	0.00									
10	920114	420.44	414.32	420.44	6.12	5.08	0.00									
11	920115	421.18	418.79	420.77	2.39	0.74	0.00									
12	920116	420.85	415.37	418.21	5.48	0.00	3.42									
13	920117	419.45	416.00	418.86	3.45	0.00	0.00									
14	920120	418.86	415.80	416.36	3.06	0.00	0.20	TR14	+DM14	-DM14	+DI14	-DI14	DI	DI	DX	ADX
15	920121	416.39	411.32	412.64	5.07	0.00	4.48									
16	920122	418.13	412.49	418.13	5.64	1.74	0.00	56.11	12.62	12.36	22	22	0	44	0	
17	920123	419.78	414.36	414.96	5.42	1.65	0.00	57.52	13.37	11.48	23	19	4	42	9	
18	920124	417.27	414.29	415.48	2.98	0.00	0.07	56.39	12.41	10.73	22	19	3	41	7	
19	920127	416.84	414.48	414.99	2.36	0.00	0.00	54.72	11.53	9.96	21	18	3	39	7	
20	920128	416.41	414.54	414.96	1.87	0.00	0.00	52.69	10.70	9.25	20	17	3	37	8	
21	920129	417.83	409.17	410.34	8.66	0.00	5.37	57.58	9.94	13.96	17	24	7	41	17	
22	920130	412.17	409.26	411.62	2.91	0.00	0.00	56.38	9.23	12.96	16	22	6	38	15	
23	920131	412.63	408.64	408.78	3.99	0.00	0.62	56.34	8.57	12.65	15	22	7	37	18	
24	920203	409.95	407.45	409.53	2.50	0.00	1.19	54.82	7.96	12.94	14	23	9	37	24	
25	920204	413.85	409.28	413.85	4.57	3.90	0.00	55.47	11.29	12.02	20	21	1	41	2	
26	920205	416.17	413.18	413.84	2.99	2.32	0.00	54.50	12.80	11.16	23	20	3	43	6	
27	920206	414.55	411.93	413.82	2.62	0.00	1.28	53.22	11.88	11.61	22	21	1	43	2	
28	920207	415.29	408.04	411.09	7.25	0.00	3.89	56.67	11.04	14.67	19	25	6	44	13	10
29	920210	413.77	411.07	413.77	2.70	0.00	0.00	55.32	10.25	13.62	18	24	6	42	14	10
30	920211	414.38	412.24	413.76	2.14	0.61	0.00	53.51	10.12	12.65	18	23	5	41	12	10

SIDEBAR FIGURE 2: ADX SPREADSHEET. Here is the output for the S&P 500 from January 1 to February 11, 1992.

The smoothing formula for the TR14 column begins at cell H17:

$$=Round((TRUNC((H16-(H16/14)+E17),3),2)$$

The smoothing formula subtracts 1/14th of yesterday's TR14 from yesterday's TR14 and then adds today's TR value. The rounding((truncating function is used to calculate the indicator as close as possible to the developer of the ADX's original form of calculation (which was done by hand).

The smoothing formula for the +DM14 column begins at cell I17:

=Round((TRUNC((I16-I16/14)+F17),3),2)

The smoothing formula subtracts 1/14th of yesterday's +DM14 from yesterday's +DM14 and then adds today's +DM value. The smoothing formula for the -DM14 column begins at cell J17:

=Round((TRUNC((J16-(J16/14)+G17),3),2)

The smoothing formula subtracts 1/14th of yesterday's -DM14 value from yesterday's -DM14 and then adds today's -DM value. Now we have a 14-day smoothed sum of TR, +DM and -DM. The next step is to calculate the ratios of +DM and -DM to TR. The ratios are called the +directional indicator (+DI) and -directional indicator (-DI). The formula for the +DI column begins at cell K16:

=Round((100*(I16/H16)),0)

The formula for the +DI column begins at cell L16:

=Round((100*(J16/H16)),0)

The next step is to calculate the absolute value of the difference between the +DI and the -DI. This is done in column M and the formula for cell M16:

=ABS(K16-L16)

The next column calculates the sum of the +DI and -DI. The formula for cell N16:

=K16+L16

The next step is to calculate the Dx, which is the ratio of the absolute value of the difference between the +DI and the -DI divided by the sum of the +DI and the -DI. This is done in column O. The formula for cell O16:

=Round(100*(M16/N16)),0)

The final step is smoothing the Dx to arrive at the value of the ADX. First, average the last 14 days of Dx values. The formula for cell P28:

=AVERAGE(O15:O28)

The smoothing process uses yesterday's ADX value multiplied by 13, and then add today's Dx value. Finally, divide this sum by 14. The formula for cell P29:

=Round((((P28*13)+O29)/14),0)

—Thom Hartle, Editor